Datumsabfragen im Klartext, Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

In Ihren Datenbanken müssen Sie regelmäßig Auswertungen erstellen, die Ihnen beispielsweise Auskunft darüber geben, welche Bestellungen an einem Mittwoch aufgegeben wurden oder welche Arbeitszeiten an einem Samstag angefallen sind. Normalerweise erstellen Sie dazu pro Wochentag eine Abfrage wie "Bestellungen/Mittwochs" und verwenden zum Beispiel ein Kriterium wie das folgende:

Weekday([Bestelldatum]) = 4

Die Funktion "WeekDay()" liefert für das als Parameter übergebene Datum als Ergebnis eine Zahl von "1" bis "7", wobei "1" für "Sonntag", "2" für "Montag", "3" für "Dienstag" usw. steht. Das gezeigte Kriterium sorgt also dafür, dass nur die Datensätze selektiert werden, bei denen die Funktion "WeekDay()" als Ergebnis "4" für "Mittwoch" liefert.

Statt der hierbei erforderlichen sieben Abfragen wäre eine Parameterabfrage, die den gewünschten Wochentag vor Ausführung der Abfrage für einen Parameter "Wochentag" vom Typ "Integer" erfragt, wesentlich praktischer. Das Kriterium würde dann beispielsweise so aussehen:

Weekday([Bestelldatum]) = [Wochentag]

Das lässt sich zwar realisieren, aber Sie müssten im Dialog "Parameterwerte eingeben" eine Zahl zwischen "1" und "7" eingeben, also immer im Kopf haben, dass "1" dem Tag "Sonntag", "2" dem Tag "Montag", "3" dem Tag "Dienstag" und so weiter entspricht. Eine komfortablere Abfrage würde die Eingabe des betreffenden Tages im Klartext, also als "Montag", "Mittwoch" oder "Samstag" erlauben. Mit Hilfe einer globalen Funktion lässt sich das wie folgt realisieren:

· Kopieren Sie die folgende Funktion in ein Modul:

Function WeekdayStr(varDate As Variant) As String
  Dim intWD As Integer

  On Error Resume Next
  If Not IsDate(varDate) Or IsNull(varDate) Then
    WeekdayStr = "???"
  Else
    intWD = WeekDay(CDate(varDate))
    WeekdayStr = _
      Choose(intWD, "Sonntag", "Montag", _
                    "Dienstag", "Mittwoch", _
                    "Donnerstag", "Freitag", _
                    "Samstag")
  End If

End Function
· Öffnen Sie die gewünschte Abfrage im Entwurfsmodus.

· Wählen Sie das Menü Abfrage-Parameter an, richten Sie einen Parameter "Wochentag" vom Typ "Text" ein und klicken Sie auf Ok.

· Geben Sie als Kriterium für das abzufragende Feld, also zum Beispiel für "Bestelldatum", den folgenden Ausdruck ein:

WeekdayStr([Bestelldatum])=[Wochentag]

Für "Bestelldatum" geben Sie den Namen des Feldes an, das in Ihrer Abfrage geprüft werden soll.

· Speichern Sie die Änderungen und lassen Sie die Abfrage einmal ausführen.

Access zeigt den Dialog "Parameterwerte eingeben" an, in dem Sie nun den gewünschten Tag im Klartext eingeben können. Anschließend wird für jeden Datensatz die Funktion "WeekdayStr()" aufgerufen, die als Ergebnis den Klartext des Wochentages für das als Parameter übergebene Datum liefert, den Access dann gegen den eingegeben Parameter prüft und bei Übereinstimmung den Datensatz selektiert.

Wenn Sie mehrere Wochentage auf einmal, also beispielsweise Tätigkeiten für Samstag UND Sonntag abfragen möchten, verwenden Sie folgendes Kriterium in der Abfrage:

InStr([Wochentag];WeekdayStr([Tätigkeitsdatum]))<>0

Im Dialog "Parameterwerte eingeben" geben Sie dann "Samstag Sonntag" ein. Anschließend wird für jeden Datensatz die Funktion "WeekdayStr()" aufgerufen, die als Ergebnis den Klartext des Wochentages für das als Parameter übergebene Datum liefert. Mit Hilfe der Funktion "InStr()" wird dann geprüft, ob dieser Wochentag in der Eingabe enthalten, das Ergebnis also ungleich "0" ist.

Im zweiten Teil dieses Tipps der Woche stellen wir eine alternative Lösung vor, die Ihnen die Eingabe von "Woche", "Monat", "Quartal" oder "Halbjahr" erlaubt, so dass Sie bei Auswertungen für diese Zeiträume nicht erst im Kalender nachschlagen müssen.

Im ersten Teil dieses Tipps der Woche hatten wir Ihnen eine Lösung vorgestellt, die die Eingabe von Wochentagen wie "Montag", "Mittwoch" oder "Samstag Sonntag" für ein Abfragekriterium im Klartext erlaubt. Mit einer vergleichbaren Lösung können Sie auch Auswertungen für bestimmte Zeiträume wie "Diese Woche", "Diesen Monat", "Dieses Quartal" oder "Dieses Halbjahr" vereinfachen.

Basis dieser Lösung ist eine globale Funktion "CheckDate(<Datum>,<Zeitraum>)", die für das als ersten Parameter übergebene Datum prüft, ob es in dem über den zweiten Parameter spezifizierten Zeitraum liegt. Ist das der Fall, liefert die Funktion das Ergebnis "True/Wahr", andernfalls "False/Falsch". Den Zeitraum legen Sie über einen Abfrageparameter "Zeitraum" vom Datentyp "Text" fest und geben ihn im Dialog "Parameterwerte eingeben" mit "Woche", "Monat", "Quartal" oder "Halbjahr" vor der Ausführung der Abfrage ein. Ein Kriterium wie das folgende sorgt dann dafür, dass nur die Datensätze selektiert werden, bei denen zum Beispiel ein Bestelldatum im gewünschten Zeitraum liegt:

CheckDate([Bestelldatum];[Zeitraum]) = Wahr

Um diese Lösung in Ihre Datenbank einzubauen, gehen Sie wie folgt vor:

· Kopieren Sie die folgende Funktion in ein neues oder ein vorhandenes Modul:

Function CheckDate(varDate As Variant, _
                   strRange As String) As Boolean
  Dim intKWNow As Integer, intKWCheck As Integer
  Dim intMNow As Integer, intMCheck As Integer
  Dim intQNow As Integer, intQCheck As Integer
  Dim intHYNow As Integer, intHYCheck As Integer

  On Error Resume Next
  CheckDate = False 'Default: Außerhalb des Zeitraums
  If Not IsDate(varDate) Or _
         IsNull(varDate) Then Exit Function
  If Year(Now) <> Year(varDate) Then Exit Function

  Select Case strRange
    Case "Woche"
      intKWNow = DatePart("ww", Now, vbMonday, _
                 vbFirstFourDays)
      intKWCheck = DatePart("ww", varDate, vbMonday, _
                   vbFirstFourDays)
      CheckDate = (intKWNow = intKWCheck)
    Case "Monat"
      intMNow = Month(Now)
      intMCheck = Month(varDate)
      CheckDate = (intMNow = intMCheck)
    Case "Quartal"
      intQNow = DatePart("q", Now, vbMonday, _
                vbFirstFourDays)
      intQCheck = DatePart("q", varDate, vbMonday, _
                  vbFirstFourDays)
      CheckDate = (intQNow = intQCheck)
    Case "Halbjahr"
      intHYNow = IIf(Month(Now) < 7, 1, 2)
      intHYCheck = IIf(Month(varDate) < 7, 1, 2)
      CheckDate = (intHYNow = intHYCheck)
  End Select

End Function
· Öffnen Sie die betreffende Abfrage im Entwurfsmodus.

· Wählen Sie das Menü Abfrage-Parameter an, richten Sie einen Parameter "Zeitraum" vom Typ "Text" ein und klicken Sie auf Ok.

· Geben Sie als Kriterium für das abzufragende Feld, wie das "Bestelldatum", den folgenden Ausdruck ein:

CheckDate([Bestelldatum];[Zeitraum]) = Wahr

Für "Bestelldatum" geben Sie den Namen des Feldes an, das in Ihrer Abfrage geprüft werden soll.

· Speichern Sie die Änderungen und lassen Sie die Abfrage einmal ausführen.

Access zeigt den Dialog "Parameterwerte eingeben" an, in dem Sie nun den gewünschten Zeitraum mit "Woche", "Monat", "Quartal" oder "Halbjahr" definieren. Anschließend wird für jeden Datensatz die Funktion "CheckDate()" aufgerufen. Die Funktion ermittelt zunächst je nach Zeitraum die aktuelle Kalenderwoche, den aktuellen Monat, das aktuelle Quartal oder das aktuelle Halbjahr ausgehend vom Systemdatum "Now". Danach wird die Kalenderwoche, der Monat, das Quartal oder das Halbjahr für das als ersten Parameter übergebene Datum ermittelt. Ein Vergleich der Form

(intKWNow = intKWCheck)

liefert "True/Wahr" oder "False/Falsch" und somit das Funktionsergebnis. Abhängig davon wird dann der aktuelle Datensatz selektiert oder nicht.

